Tutorial : 12 SCILAB Programming – Applications of Chemical Engineering Problems (Unsteady state problems)

Date : 6/10/2016

Aim: To obtain solution for Ordinary differential equation of first order in reaction kinetics using SCILAB

Problem Statements

Exercise 1:

Consider four chemical species *A*, *B*, *C* and *D* that represents O_2, H_2, OH, H_2O. The reaction between these species can be written as

$$A + B \overset{k_1}{\rightarrow} C + D \qquad (1)$$

This reaction is not an instantaneous reaction, but will react at a rate proportional to the concentration of reactants A and B with rate constant k_1. Consider that there is a second reaction and the reaction is represented as

$$A + A + C \overset{k_2}{\rightarrow} B \qquad (2)$$

The above two reaction takes place in a reaction system. From the reaction scheme the concentration of chemical species are as follows:

$$\frac{d[A]}{dt} = -k_1[A][B] - 2k_2[A][A][C]$$

$$\frac{d[B]}{dt} = -k_1[A][B] + k_2[A][A][C]$$

$$\frac{d[C]}{dt} = k_1[A][B] - k_2[A][A][C]$$

$$\frac{d[D]}{dt} = k_1[A][B]$$

In the above equations square brackets denotes the concentration of species *A*, *B*, *C*, and *D*. Let us see the first equation

$$\frac{d[A]}{dt} = -k_1[A][B] - 2k_2[A][A][C]$$

The rate of reaction (1) will depend proportionally on [*A*] and [*B*], the constant of proportionality given by k_1. So the rate of reaction 1 will be given by $k_1[A][B]$. Each instance of reaction 1 will use up one unit of A. So the rate of change of concentration of A due to reaction 1 will be equal to $-k_1[A][B]$. The minus signifying one unit being used up. There are similar terms in the other ODE's corresponding to using up the species B and creating species C and D. The rate of the second reaction will depend proportionally on $[A]$, $[A]$ again, and $[C]$ and so the rate of reaction (2) will be $k_2[A][A][C]$. Each reaction (2) will use up 2 units of A, so the rate of change of concentration of A due to reaction 2 will be equal to $-2k_1[A][A][C]$, with similar terms in the ODE's denoting using one unit of C and creating one unit of B.Use Scilab to approximate the solution to this system of ODE's. Consider, t = 0; 0.001; 0.002; ….0.1. Take k_1 = 1e2, k_2 = 1, Initial concentrations are $[A]$ =1 mol/L, $[B]$ = 1mol/L, $[C]$ = 0 mol/L, Also plot the graph showing the concentration profile.

SCILAB Code:

Now we can use SCILAB to approximate the solution to this system of ODE's. We have four chemical species, so we will have a system of four equations. We will make the relationship c_1 = [A], c_2 = [B], c_3 = [C], c_4 = [D].

Open SCINOTES and create the following function file:

```
function dc=rate(t, c)
    f1 = k1*c(1)*c(2)
    f2 = k2*c(1)*c(1)*c(3)
    dc(1) = -f1 - 2*f2
    dc(2) = -f1 + f2
    dc(3) = f1 - f2
    dc(4) = f1
endfunction
```

Now goto console window and execute the function file

```
-->exec('C:\Users\User-Pc\Desktop\rate.sci', -1)
-->k1 = 1e2;
-->k2 = 1;
-->t = 0:0.001:0.1;
-->c0 = [1 ;1; 0; 0];
-->c = ode(c0,0,t,rate);
-->plot(t,c)
-->xgrid(1)
-->title('Rate Kinetics ODE')
-->xlabel('Time,min','fontsize',4)
-->ylabel('Concentration, mol/L','fontsize',4)
-->legend(['A','B','C','D'],[1 2 3 4],"ur")
```

Concentration Profile



Exercise 2: The dynamic model for an isothermal, constant volume, chemical reactor with asingle second order reaction is:

$$\frac{dC_A}{dt} = \frac{F}{V}C_{Af} - \frac{F}{V}C_A - kC_A^2$$

$$\frac{F}{V} = 1\min^{-1}, C_{Af} = 1 gmol/liter, k = 1 liter/gmol.\min$$

Find the steady-state $f(x) = -x2$ and substituting the parameter and input values we find

$$1 - C_{As} - C_{As}^2 = 0$$

where the subscript s is used to denote the steady-state solution. For notational convenience, let $x = C_{As}$ and write the algebraic equation as

$$f(x) = -x^2 - x + 1 = 0$$

We can directly solve this equation using the quadratic formula to find x.

Use SCILAB and determine the value of x.

Solution: Goto console window and try the following:

```
-->x=poly(0,'x');
-->p=-x^2-x+1
 p  =
             2
    1 - x - x
 -->roots(p)
```

```
ans  =
 - 1.618034
   0.6180340
```

Therefore, x = -1.618 and x = +0.618 to be the solutions. Obviously a concentration cannot be negative, to the only physically meaningful solution is x = 0.618.

Result

Thus we learned the solution for ODE's in reaction kinetics using SCILAB